M_E_M_O

DATE: 08 November 2002

TO: R. L. Tregoning and M. T. Kirk

FROM: P. T. Williams and B. R. Bass

SUBJECT: Status Report on Davis-Besse Analyses

The attached Figs. 1-9 provide a summary of the Davis-Besse analyses performed to date under the new Task 9 of JCN Y6533. In Fig. 1, the cladding properties used in the current study are presented: (a) true stress versus true strain and (b) thermal expansion coefficient versus temperature. The remaining figures address a specific sub-task described in the workscope for Task 9.

Sub-task 9.1D requires an estimate for crack driving forces as a function of flaw size and applied membrane stress in cladding. Table 1 shows the Case Matrix developed for this subtask.

Figure 2 depicts the first step carried out in preparation for the *J*-integral analyses, i.e., calculation of an updated estimate of the exposed cladding "footprint" based on the recent "dental mold" cast from the D-B cavity. That footprint area was estimated to be 28.23 in^2. Comparisons of the latest "footprint" statistics with previous ORNL interpretations are given in the table of Fig. 2(b). The newly calculated "footprint" area was used to define a burst disk having the same lateral surface area under load.

Table 2 presents ductile tearing data for three-wire series-arc stainless steel weld overlay cladding published in NUREG/CR-5511 [1]. Table 3 presents additional ductile tearing data for stainless steel cladding [2]. The ductile-tearing data in Tables 2 and 3 are plotted as a function of temperature in Fig. 3. The J_{Ic} data at 288 °C (550 °F) has been extrapolated to 318.33 °C (605 °F) using a curve fit developed from Table 2. The extrapolated J_{Ic} data are also shown in Table 3.

The *ExpertFit*© statistical software [3] was used to fit a statistical distribution to the data in Table 3 for 318.33 °C (605 °F). The resulting fit is a log-logistic cumulative distribution of the form

$$CDF = F(J_{Ic},\alpha,\beta,\gamma) = \frac{1}{1+\left[\dfrac{(J_{Ic}-\gamma)}{\beta}\right]^{-\alpha}} \tag{1}$$

(2)

where $\alpha = 9.12897$, $\beta = 79.46842$ kJ/m^2, and $\gamma = 0$. The corresponding percentile function is

$$J_{Ic}(P,\alpha,\beta,\gamma) = \gamma + \beta \exp\left[\frac{-\ln\left(\dfrac{1-P}{P}\right)}{\alpha}\right];\ (0 < P < 1) \tag{3}$$

Figure 4 shows a density/histogram overplot and the *CDF* of the log-logistic distribution.

H-44

Figure 5 presents three finite-element models developed for this phase of the analysis. Surface-breaking flaws were centrally located in each burst disk with the three relative flaw depths: a/t = 0.5, 0.25, and 0.05 for a flaw length of 2.0 inches (50.8 mm).

Each model was loaded with an increasing lateral pressure. The calculated J-integral loading paths for the three models are shown in Fig. 6a. The *TableCurve® 3D* [4] software was then used to fit a surface of critical pressure for ductile tearing as a function of J_{Ic} and relative flaw depth, a/t, (see Fig. 6b). The resulting surface was

$$\begin{aligned}\ln(P_{DT}) = -2.409372112829741 - 0.9619997646244488[\ln(a/t)] - \\ 1.674757970053341\times10^{-6}\left[J_{Ic}^2\right] + 0.8312894992181433[\ln(J_{Ic})]\end{aligned} \quad (4)$$

where P_{DT} is MPa and J_{Ic} is in kJ/m^2. Equation (4) can be interpreted as given a value for relative flaw depth ($2L$ = 2.0 inches) and J_{Ic} for the cladding, P_{DT} is the lateral pressure required for the onset of ductile tearing of the flaw in the burst disk (radius = 3.0 inches, thickness = 0.25 inches).

An estimated cumulative distribution function for relative flaw depths in cladding [5] is shown in Fig. 7a. The *CDF* of Fig. 7a was sampled (N = 1000) to produce the histogram shown in Fig. 7b.

The plastic collapse curve shown in Fig. 8a was developed using the models in Fig. 5. The curve in Fig. 8a represents the *pressure at numerical instability*, P_{NI}, that can be applied to the scaled log-Laplace statistical distribution described in [6].

$$Q_{LP}(p\,|\,0, 1.1057\times P_{NI}, 11.45441) = \quad (5)$$

$$BP_p = \begin{cases} \exp\left[\ln(1.1057\times P_{NI}) + \dfrac{\ln(2p)}{11.45441}\right] & ; p \le 0.5 \\ \exp\left[\ln(1.1057\times P_{NI}) - \dfrac{\ln[2(1-p)]}{11.45441}\right] & ; p > 0.5 \end{cases} \quad \text{for } (0 < p < 1)$$

where from Fig. 8a,

$$P_{NI} = 52.739 - 39.656(a/t) - 57.659(a/t)^2 \quad (6)$$

A Monte Carlo code (see Exhibit 1 for a listing) was developed to simulate a large number of possible flaw configurations (single flaw located in the middle of the representative burst disk) with sampled relative flaw depths (flaw length = 2.0 inches for all realizations, see Fig. 7), sampled uncertainties in plastic collapse burst pressure (Eqs. (5) and (6)), and sampled uncertainties in the required pressure for the onset of ductile tearing (Eqs. (3) and (4). The results from 5000 simulations are shown in Fig. 9. From Fig. 9, it is clear that, for the relative flaw depth *CDF* assumed in the analysis, plastic collapse is the dominant failure mode compared to the onset of ductile tearing with a cumulative probability for plastic collapse of 0.996 at a median pressure of 8.4 ksi. Additional calculations will be carried out to check the sensitivity of the results to the number of simulations used in the analysis.

References

[1] F. M. Haggag, W. R. Corwin, and R. K. Nanstad, *Irradiation Effects on Strength and Toughness of Three-Wire Series-Arc Stainless Steel Weld Overlay Cladding,* NUREG/CR-5511 (ORNL/TM-11439), Oak Ridge National Laboratory, February 1990.

[2] Personal communication from R. L. Tregoning, November 4, 2002.

[3] A. M. Law, *ExpertFit© User's Guide*, Averill M. Law and Associates, 2002.

[4] *TableCurve 3D – Automated Surface Fitting and Equation Discover*, Version 3.0 for Windows NT, *User's Manual,* SPSS Software, 1997.

[5] Personal communication from R. L. Tregoning, November 4, 2002.

[6] P. T. Williams and B. R. Bass, "Stochastic Failure Model for the Davis-Besse RPV Head," USNRC Letter Report ORNL/NRC/LTR-02/10 (under review).

Table 1. Case Matrix for Task 9.1D

Case Number	*a* (inches)	2*L* (inches)	*a/t* (-)	2*L/a* (-)
9.1D1	0.1250	2	0.50	16
9.1D2	0.0625	2	0.25	32
9.1D3	0.0125	2	0.05	160
9.1D4	0.1250	1	0.50	8
9.1D5	0.0625	1	0.25	16
9.1D6	0.0125	1	0.05	80
9.1D7	0.1250	0.375	0.50	3
9.1D8	0.0625	0.375	0.25	6
9.1D9	0.0125	0.375	0.05	30

Table 2. Ductile Tearing Data Extracted from Table 13 of NUREG/CR-5511.

Specimen	Test Temperature (°C)	J_{Ic} (kJ/m^2)	Tearing Modulus
	Unirradiated Specimens		
A13G	-75	117	64
H2	-75	137	49
A15B[a]	20	165	270
A13D	20	134	209
A10G	20	171	176
A10E	120	128	246
H5	120	119	229
H3	120	120	232
A13F[a]	120	159	359
H6	200	90	240
H4	200	111	231
A15D	288	77	267
A13C	288	66	170
H1	288	82	192
	Irradiated Specimens		
A15F	-75	78	40
A15G	-75	56	36
A13A	30	144	177
A15C	50	124	146
A10F	120	94	175
A15A	288	25	191

[a] Specimen was not side-grooved, while all other specimens in table were side-grooved 20%.

Table 3. Ductile Tearing Data Used in Development of J_{Ic} Statistical Distribution

J_{IC} Distribution Values.							
Specimen ID	Test Temp (°C)	J_{Ic} kJ/m²	Reference	Remarks		Op. Temp (°C)	J_{Ic} kJ/m²
A15D	288	78	NUREG/CR-5511	Three-Wire Cladding Study	1	318 33	74 07
A13C	288	68	NUREG/CR-5511	Three-Wire Cladding Study	2	318 33	64 57
H1	288	79	NUREG/CR-5511	Three-Wire Cladding Study	3	318 33	75 01
H10	288	85	NUREG/CR-6363	Aged 3-wire cladding @ 288C for 1605 hours	4	318 33	80 71
AA04	288	93	NUREG/CR-6363	Aged 3-wire cladding @ 288C for 1605 hours	5	318 33	88 31
AA02	288	59	NUREG/CR-6363	Aged 3-wire cladding @ 288C for 1605 hours	6	318 33	56 02
AA13	288	91	NUREG/CR-6363	Aged 3-wire cladding @ 288C for 20,000 hours	7	318 33	86 41
AA15	288	77	NUREG/CR-6363	Aged 3-wire cladding @ 288C for 20,000 hours	8	318 33	73 12
H15	288	111	NUREG/CR-6363	Aged 3-wire cladding @ 343C for 20,000 hours	9	318 33	105 40
H16	288	110	NUREG/CR-6363	Aged 3-wire cladding @ 343C for 20,000 hours	10	318 33	104 45

Data extrapolated to 318.33 °C (605 °F).



Fig. 1. Cladding properties used in the current study: (a) true stress vs true strain and (b) thermal expansion coefficient.



(a)

Description	Scaling Factor	Area	Perimeter	Centroid of Wastage Area Footprint x_c	y_c	Moments of Interia About the Centroid I_{xx}	I_{yy}	I_{xy}	Eigenvalue Extraction for Prinicipal Moments and Directions: Principal Moments I_1	I_2	Principal Directions I_1	I_2
		(in²)	(in.)	(in.)	(in.)	(in⁴)	(in⁴)	(in⁴)	(in⁴)	(in⁴)	$<n_x, n_y>$	$<n_x, n_y>$
As-Found Footprint	1	35 36	30 36	16 4122	-0 1194	98.89	9699 33	-117 16	75.26	197.41	<0 9004, -0 4351>	<0 4351, 0 9004>
Adjusted Footprint for Bounding Calculation	0.25 in	40 06	31 78	16 4301	-0 1255	129 02	11031 81	-141.35	99 00	245 71	<0 8943, -0 4476>	<0 4476, 0 8943>
As-Found Footprint 9/23/2002	1	28 23	24 55	15 332	-0 18	95.56	6708 63	-50 52	54 01	113 07	[0 558 0 830]	[-0 830 0 558]

Footprint centroid is in global coordinates.
Global coordinate system has its z-axis aligned with the vertical centerline of the vessel
The x-y plane of the global coordinate system is a horizontal plane
(b) with the x-axis along the line between the centerlines of Nozzles 3 and 11

Fig. 2. Latest footprint estimated from "dental mold".



Fig. 3. Ductile tearing data for stainless steel weld overlay cladding: (a) J_{Ic} data from unirradiated specimens and (b) tearing modulus data from unirradiated specimens



(a)



(b)

Fig. 4. Statistical distribution for J_{Ic} at 318.33 °C (605 °F): (a) histogram of data with fitted log-logistic density overplot and (b) log-logistic cumulative distribution function compared to cumulative probabilities of J_{Ic} data estimated by median rank order statistic $p = (i-0.3)/(n+0.4)$.



Fig. 5. Finite-element models used in calculating applied *J*-integrals produced by pressure loading of burst disk: (a) Model 9.1D1 ($a/t = 0.5$, $2L/a = 16$) (b) Model 9.1D2 ($a/t = 0.25$, $2L/a = 32$), and (c) Model 9.1D3 ($a/t = 0.05$, $2L/a = 160$) (Task 9.1D)



Fig. 6. Surface fit of *J-integral* driving forces – applied pressure as a function of J_I and *a/t* for a 2-inch long flaw : (a) driving forces, (b) fitted surface.



Fig. 7. Relative flaw depth cumulative distribution function: (a) curve fit to CDF and (b) histogram resulting from random sampling (1000 sampled flaws) of CDF.



Fig. 8. (a) Plastic collapse curve of *pressure at numerical instability*, _P_{NI}, as a function of *a/t* for a 2.0 inch flaw centered in the representative burst disk and (b) Log-Laplace CDF generated from previous study.



Fig. 9. Results of Monte Carlo sampling of 5000 flaws (2.0 inches in length) comparing (a) plastic collapse critical pressures (burst pressures), (b) onset of ductile tearing critical pressures, and (c) ratio of plastic collapse to ductile tearing critical pressures for each simulation.

Exhibit 1. Listing of Monte Carlo Code Simulating Flaws in Cladding

```
      program monte_carlo
C==================================================================
      implicit none
C==================================================================
      integer                           :: iseed, i
      double precision                  :: P1, P2, P3, BP, PDT
      double precision                  :: JIc, at, PNI
      double precision, parameter       :: one=1.
      integer, parameter                :: Ntrials=5000
      double precision,dimension(Ntrials):: P_Order
      double precision,dimension(Ntrials):: BPI, PDTI, JIcI
      double precision,dimension(Ntrials):: atI, Ratio, RB
      integer,dimension(Ntrials)        :: iperm1,iperm2,iperm3
      integer,dimension(Ntrials)        :: iperm4,iperm5
C==================================================================
c     rnset  - set random number generator see
c     drnunf - random number generator
c     dsvrgp - sort vector in ascending order
C==================================================================
      external RNSET                              ! IMSL Routines
      double precision, external :: drnunf, DSVRGP  ! IMSL Routines
C==================================================================
      open (unit=10,file='Monte.DAT',status='UNKNOWN')
      iseed = 123457
      call RNSET(iseed) ! IMSL Routine: Set seed
      write (10,1000)
C==================================================================
      do i=1,Ntrials
C==================================================================
        iperm1(i) = i
        iperm2(i) = i
        iperm3(i) = i
        iperm4(i) = i
        iperm5(i) = i
C==================================================================
        P_Order(i) = (REAL(i)-0.3d0)/(REAL(Ntrials)+0.4d0)
C==================================================================
        P1 = DRNUNF() ! IMSL Function: Random Number Generator
        P2 = DRNUNF() ! IMSL Function: Random Number Generator
        P3 = DRNUNF() ! IMSL Function: Random Number Generator
C==================================================================
        if ( P1 <= 0.99585d0 ) then
          at = 0.0167d0*P1
        else
          at = 62294.0d0 - 125045.0d0*P1 + 62752.0d0*P1**2
        endif
C=======================================================
C       Plastic Collapse
C=======================================================
        PNI = 52.739d0 - 39.656d0*at - 57.659d0*(at**2)
        PNI = MAX(PNI,0.0001d0)
        if ( P2 .LE. 0.5d0 ) then
          BP = exp( LOG(1.1057d0*PNI) + (LOG(2.0d0*P2)/
     &                                  11.45441d0) )
        else
          BP = exp( LOG(1.1057d0*PNI) - (LOG(2.0d0*(one-P2))/
     &                                  11.45441d0) )
        endif
C=======================================================
C      Ductile Tearing
C=======================================================
```

```
        JIc = 79.46482d0*EXP( -LOG( (one-P3)/P3 )/9.12897d0 )
        CALL Driving_Force(at,JIc,PDT)
C==============================================================
        BPI(i)   = BP/6.894757d0
        PDTI(i)  = PDT/6.894757d0
        Ratio(i) = BPI(i)/PDTI(i)
        atI(i)   = at
        JIcI(i)  = JIc
      enddo
C==============================================================
C      Ductile Tearing
C==============================================================
C      Sort by increasing magnitude
C==============================================================
      CALL DSVRGP(Ntrials,atI,  RB,iperm1)
      CALL DSVRGP(Ntrials,BPI,  RB,iperm2)
      CALL DSVRGP(Ntrials,PDTI, RB,iperm3)
      CALL DSVRGP(Ntrials,Ratio,RB,iperm4)
      CALL DSVRGP(Ntrials, JIcI,RB,iperm5)
C==============================================================
      DO i=1,Ntrials
        write (10,1100) atI(iperm1(i)),BPI(iperm2(i)),
     &                  PDTI(iperm3(i)),Ratio(iperm4(i)),
     &                  JIcI(iperm5(i)),P_Order(i)
      ENDDO
C==============================================================
C
1000  FORMAT('    a/t          BP(ksi)       PDT(ksi)',
     &       '        BP/PDT        JIc         P_Order')
1100  FORMAT(6ES14.6)
      end program monte_carlo
*-----------------------------------------------------------*
      SUBROUTINE Driving_Force(at,JI,Press)
*-----------------------------------------------------------*
* TableCurve 3D
* X = a/t
* Y = JI
* Z = Press
* Eqn#= 151240781
* Eqn   =  lnz = a + b*ln(x) + c*y^2 + d*ln(y)
* r2    =  0.9901929705470828
* r2adj =  0.990041510632752
* StdErr=  1.443354380463457
* Fstat = 8750.531904291727
* a     = -2.409372112829741
* b     = -0.9619997646244488
* c     = -1.674757970053341E-06
* d     =  0.8312894992181433
*-----------------------------------------------------------*
      DOUBLE PRECISION at,JI,Press
      DOUBLE PRECISION x,y,z
      DOUBLE PRECISION f1,f2,f3
*-----------------------------------------------------------*
      x  = MAX(at,0.00000001d0)
      y  = MAX(JI,0.00000001d0)
*-----------------------------------------------------------*
      f1 = LOG(x)
      f2 = y*y
      f3 = LOG(y)
*-----------------------------------------------------------*
      z  = -2.409372112829741D0-0.9619997646244488D0*f1
     &     -1.674757970053341D-06*f2+0.8312894992181433D0*f3
      z   = EXP(z)
      Press = z
```

```
*----------------------------------------------------------*
      RETURN
      END
```